



REXX Tools Panel

Charles Rice

crice@cprsystems.com



REXX Tools

- Batch Console
- Network Monitor
- Email Reports

```

CATALOG VSECONX.PROC                                REPLACE=YES
/* ----- */
/* ISSUE MAP COMMAND FOR STATIC PARTITIONS          */
/* ----- */
FC=SYSVAR('SYSPID')
CONNAME='VSECON'||SYSPID                          /* MAKE CONSOLE NAME UNIQUE */
ADDRESS CONSOLE
COM1='ACTIVATE NAME' CONNAME 'PROFILE REXNORC'
COM1                                               /* ACTIVATE THE CONSOLE */
'MAP'
RC=GETMSG(MSG.,'RESP',,,5)
/* ----- */
/* BUILD PARTITION TABLE FOR STATIC PARTITIONS     */
/* ----- */
PART.0=0
DO I=4 TO 15
  PARSE VAR MSG.I F1 F2 F3 F4 F5 F6 F7 F8 F9 F10 F11
/* MOVE FIELDS FOR MULTIPLE PARTITIONS IN SAME ADDRESS SPACE */
  IF F4 = "*" THEN DO
    F4 = F5
    F5 = F6
    F6 = F7
    F7 = F8

```

```

    F8 = F9
    F9 = F10
    F10 = F11
END
/* ADD TO PARTITION TABLE IF ACTIVE JOB IN PARTITION */
IF F10 \= " " THEN DO
    J=PART.0+1
    PART.J = F4
    JOB.J = F10
    PART.0 = J
    JOB.0=J
END
END
/* ----- */
/* GET LIST OF ALL DYNAMIC CLASSSES */
/* ----- */
'MAP CLASS=ALL'
RC=GETMSG(MSG., 'RESP' , , , 5)
CLASS.0=0
DO I = 2 TO MSG.0-1
    PARSE VAR MSG.I F1 F2 F3 F4
    J=CLASS.0+1
    CLASS.J = F3

```

```

CLASS.0 = J
END
/* ----- */
/* ADD TO PARTITION TABLE FOR ACTIVE DYNAMIC PARTITIONS */
/* ----- */
DO I = 1 TO CLASS.0
  'MAP CLASS=' || CLASS.I
  RC=GETMSG(MSG., 'RESP', , , 5)
  PARSE VAR MSG.4 F1 F2 F3 F4
  IF F3 \= "NO" THEN DO
    DO K= 6 TO MSG.0-1
      PARSE VAR MSG.K F1 F2 F3 F4 F5
      J = PART.0 + 1
      PART.J = F3
      JOB.J = F4
      PART.0 = J
      JOB.0 = J
    END
  END
END
END
/* ----- */
/* LIST PARTITION TABLE */
/* ----- */

```

```


DO I = 1 TO PART.0
  SAY PART.I JOB.I
END
/* ----- */
/* GET COMMAND FROM INPUT AND SEND COMMAND TO CONSOLE */
/* ----- */
DO FOREVER
  PULL CMD
  IF CMD="" THEN LEAVE
/* IF SLEEP COMMAND THE WAIT FOR NUMBER OF SECONDS */
  IF SUBSTR(CMD,1,5)="SLEEP" THEN DO
    CALL SLEEP STRIP(SUBSTR(CMD,7,5))
  END
  ELSE DO
/* REPLACE XXXX WITH REPLID FROM LAST MSG COMMAND */
  IF SUBSTR(CMD,1,4)="XXXX" THEN DO
    IF LENGTH(CMD) = 4 THEN CMD=PARTID
    ELSE
      CMD=PARTID||SUBSTR(CMD,5,LENGTH(CMD)-4)
    END
  IF SUBSTR(CMD,1,3)='MSG' THEN DO
    DO I = 1 TO PART.0
      IF SUBSTR(CMD,5,LENGTH(JOB.I)) = JOB.I THEN DO

```

```

        CMD=SUBSTR(CMD,1,4) || PART.I || SUBSTR(CMD,5+LENGTH(JOB.I),40)
        LEAVE
    END
END
END
IF SUBSTR(CMD,1,3)='RSP' THEN DO
    DO I = 1 TO PART.0
        IF SUBSTR(CMD,5,LENGTH(JOB.I)) = JOB.I THEN DO
            LEAVE
        END
    END
END
"REPLID"
RC=GETMSG(MSG.,'RESP',,,5)
LOC=POS(PART.I,MSG.1)
PARTID=SUBSTR(MSG.1,LOC+3,4)
CMD=PARTID || SUBSTR(CMD,5+LENGTH(JOB.I),40)
END
SAY CMD
CMD
RC = GETMSG(MSG.,'RESP',,,5)
I = 1
DO WHILE I <= MSG.0
    SAY MSG.I

```




```
I=I+1
END
IF SUBSTR(CMD,1,3) = "MSG" THEN DO
  CALL SLEEP 5
  "REPLID"
  RC=GETMSG(MSG., 'RESP' , , , 5)
  LOC=POS(SUBSTR(CMD,5,2),MSG.1)
  PARTID=SUBSTR(MSG.1,LOC+3,4)
  SAY PARTID
END
END
END
COM1='DEACTIVATE' CONNAME
COM1 /* DEACTIVATE THE CONSOLE */
EXIT
/+
```




```
// JOB VSECON
// LIBDEF PROC,SEARCH=VENDOR.REXX
// EXEC REXX=VSECONX
MSG TCPIP
XXXX SHUTDOWN
SLEEP 5
XXXX YES
RSP JCLSCHED &TERMINATE
MSG FAQSVTAM
XXXX YES
MSG FTPJCL,DATA=END
MSG CICSICCF
XXXX CEMT P SHUT,I
MSG CICSSTEST
XXXX CEMT P SHUT,I
MSG CICSPROD
XXXX CEMT P SHUT,I
Z NET,QUICK
PEND
/*
/&
```

```
CATALOG AUTOPING.PROC      EOD=/+      REPLACE=YES
/* AUTO PING TO CHECK FOR DEVICES UP */
TIMEOUT.0 = 0
J = 1
DO FOREVER
  PULL IPADD
  IF IPADD = "" THEN LEAVE
  PARSE VAR IPADD IPADD1 COMM1
  RC = SOCKET('CLIENT', 'OPEN')
  IF RC \= 0 THEN SAY 'RC='RC 'ERRMSG=' ERRMSG
  RC = SOCKET(HANDLE, 'SEND', 'PING')
  IF RC \= 0 THEN SAY 'RC='RC 'ERRMSG=' ERRMSG
  DO FOREVER
    RC = SOCKET(HANDLE, 'RECEIVE')
    IF RC \= 0 THEN SAY 'RC='RC 'ERRMSG=' ERRMSG
    IF POS('PING Ready:', BUFFER) \= 0 THEN LEAVE
    SAY BUFFER
  END
  RC = SOCKET(HANDLE, 'SEND', 'SET HOST='IPADD1)
  DO FOREVER
    RC = SOCKET(HANDLE, 'RECEIVE')
    IF RC \= 0 THEN SAY 'RC='RC 'ERRMSG=' ERRMSG
```

```
IF POS('PING Ready:',BUFFER) \= 0 THEN LEAVE
    SAY BUFFER
END
RC = SOCKET(HANDLE,'SEND','PING')
TOTMS = 0
DO I = 1 TO 5
    RC = SOCKET(HANDLE,'RECEIVE')
    IF POS('Timeout',BUFFER) \= 0 THEN DO
        SAY ' A PING REQUEST HAS TIMED OUT '
        SAY BUFFER
        TIMEOUT.J = IPADD
        TIMEOUT.0 = J
        J = J + 1
        LEAVE
    END
    SAY BUFFER
END
RC = SOCKET(HANDLE,'CLOSE')
IF RC \= 0 THEN SAY 'RC='RC 'ERRMSG=' ERRMSG
END
IF TIMEOUT.0 > 0 THEN DO
    JCL.1="* $$ JOB USER='CRR',DISP=D,CLASS=S,PRI=5,JNM=EMAIL"
```




```
JCL.2='* $$ LST CLASS=X,DISP=D,FNO=DEL,FCB=DEL '  
JCL.3='// JOB EMAIL '  
JCL.4='// LIBDEF *,SEARCH=VENDOR.TCPIP '  
JCL.5='// EXEC CLIENT,SIZE=CLIENT,PARM='APPL=EMAIL ''  
JCL.6='SET IPADDR=166.82.131.205 '  
JCL.7='SET FROM=CRICE@CAROLINASTEEL.COM '  
JCL.8='SET TO=BJOHNSON@CAROLINASTEEL.COM '  
JCL.9='SET TO=CRICE@CAROLINASTEEL.COM '  
CUR = TIME()  
IF LEFT(CUR,2) > 17 THEN DO  
    JCL.8='SET TO=BOBBY-JOHNSON@TRIAD.RR.COM '  
    JCL.9='SET TO=CRRICE@TRIAD.RR.COM '  
END  
JCL.10='SET SUBJECT=PING '  
JCL.11='TEXT '  
JCL.12='THE FOLLOWING IP ADDRESSES DID NOT RESPOND: '  
DO I = 1 TO TIMEOUT.0  
    K = 12 + I  
    JCL.K = TIMEOUT.I  
END  
K=13 + TIMEOUT.0  
JCL.K='/+'
```



```
K=14 + TIMEOUT.0
JCL.K='SEND'
K=15 + TIMEOUT.0
JCL.K='QUIT'
K=16 + TIMEOUT.0
JCL.K='/*'
K=17 + TIMEOUT.0
JCL.K='/&'
K=18 + TIMEOUT.0
JCL.K='* $$ EOJ'
JCL.0=K
ADDRESS POWER PUTQE RDR STEM JCL. WAIT 0
SAY 'JCL SUBMITTED'
END


EXIT
```



```
* $$ JOB USER='CRR',DISP=D,CLASS=S,PRI=5,JNM=AUTOPING
* $$ LST DISP=D,CLASS=X,FNO=ST1P,FCB=ST1P
// JOB AUTOPING
// LIBDEF *,SEARCH=(VENDOR.TCPIP)
// LIBDEF PROC,SEARCH=VENDOR.REXX
// EXEC REXX=AUTOPING
166.82.131.2      CSCNT
166.82.131.5      CSCNT2
166.82.131.7      CSCJAMIS
166.82.131.181    INTERMEC ACCESS POINT
166.82.131.201    SQL SERVER
166.82.131.202    CSCNT1
166.82.131.204    CSCNT4
166.82.131.205    CSCNT5
192.81.249.4      WINS SERVER
192.81.249.247    CITRIX SERVER
/*
/&
* $$ EOJ
```

```
CATALOG PRINTPDF.PROC      EOD=/+          REPLACE=YES
/* AUTO EMAIL PDF REPORTS */
JCL.0=21
JCL.1="* $$ JOB USER='CRR',DISP=D,CLASS=S,PRI=5,JNM=EMAILPDF"
JCL.2='* $$ LST CLASS=X,DISP=D,FNO=DEL,FCB=DEL '
JCL.3='// JOB EMAILPDF '
JCL.4='// LIBDEF *,SEARCH=VENDOR.TCPIP '
JCL.5="// EXEC CLIENT,SIZE=CLIENT,PARM='APPL=EMAIL '"
JCL.6='SET IPADDR=166.82.131.205 '
JCL.7='SET FROM=CRICE@CAROLINASTEEL.COM '
JCL.8='SET TO='
JCL.9='SET SUBJECT=REPORT '
JCL.10='SET PDF='
JCL.11='SET CC=ON '
JCL.12='SET DISP=DELETE '
JCL.13='TEXT '
JCL.14='ATTACHED IS YOUR REPORT IN ADOBE PDF FORMAT.'
JCL.15=' /+'
JCL.16='CD POWER.LST.A '
JCL.17='SEND '
JCL.18='QUIT '
JCL.19=' /* '
```

```
JCL.20='/&'
JCL.21='* $$ EOJ'
RESULT = OUTTRAP(PWR.)
ADDRESS POWER 'PDISPLAY LST,RJE,200,FULL=YES'
IF RC = 0 THEN
    DO I = 2 TO PWR.0 BY 2
        J=I+1
        PARSE VAR PWR.I . JNAME JNUM PRI DISP CLASS PAGES D2 FORM .
        PARSE VAR PWR.J DATE1 USER .
        DATE1=SUBSTR(DATE1,3,10)
        USER=SUBSTR(USER,4,3)
        JCL.8=LEFT(JCL.8,7)||'CRICE@CAROLINASTEEL.COM'
        IF USER='BCS' THEN JCL.8=LEFT(JCL.8,7)||'BSMITH@CAROLINASTEEL.COM'
        IF USER='BHJ' THEN
            JCL.8=LEFT(JCL.8,7)||'BJOHNSON@CAROLINASTEEL.COM'
        IF USER='DLG' THEN
            JCL.8=LEFT(JCL.8,7)||'DGAMBILL@CAROLINASTEEL.COM'
```

```
JCL.9=LEFT(JCL.9,19)||JNAME
JCL.10=LEFT(JCL.10,8)||'PDFSETUP'
IF FORM='C1P8' THEN JCL.10=LEFT(JCL.10,8)||'PDFPORT'
IF FORM='LEWW' THEN JCL.10=LEFT(JCL.10,8)||'PDFLEGAL'
JCL.16=LEFT(JCL.16,13)||CLASS
JCL.17=LEFT(JCL.17,5)||JNAME||' AS'
JCL.17=JCL.17||JNAME||'-'||DATE1||'.PDF'
IF DISP='D' THEN DO
    IF PAGES > 1 THEN DO
        ADDRESS POWER PUTQE RDR STEM JCL. WAIT 0
    END
END
END
END

EXIT
```

/+